

Appendix 3: Stata code for matching sepsis to non-hospitalized adult cohorts

Dataset in use: The dataset has one observation for each eligible person-wave.

Pertinent variables in Dataset:

unwellday=first day that a patient is hospitalized (or dies) following HRS survey (censored at 2 years from HRS survey)

hhid=household identification number for HRS

pn= "person-number" within their household (hhid & pn uniquely identify individuals within HRS)

wave= HRS survey wave

sepsrisk_percent = percentile risk of developing sepsis (propensity score)

iwdate= date of HRS interview

Dataset in storage (sepsis.dta): This dataset has one observation for each person-wave who develops sepsis as their 1st hospitalization and within 2 years after HRS survey.

Pertinent variables in dataset:

sepsis2yday = day relative to HRS survey of 1st sepsis hospitalization

Stata Code

version 13

local workhere "/Users/DIRECTORY NAME"

cd `workhere'

*make a variable for every 7th day that each patient is alive and well

```
foreach num of numlist 7 14 to 728 {  
    qui gen wellday`num'=`num' if `num'<unwellday  
}
```

*reshape the date from wide to long by groups of 1000 pts (otherwise this takes forever to run)

*reshape (observation is person-wave --> observation is person-wave-wellday)

```
foreach n of numlist 1000 2000 to 50000 {  
    preserve  
    qui keep if _n<(`n'+1) & _n>(`n'-1000)  
    qui reshape long wellday, i(hhid pn wave age sepsrisk_percent gender) j(day)  
    drop if wellday==.  
    qui save "tempobs`n'.dta", replace  
    restore  
}
```

*append re-shaped groups back together

```
use "tempobs1000.dta", clear
```

```
foreach n of numlist 2000 3000 to 50000 {  
    qui append using "tempobs`n'.dta"  
}
```

```
save "masterNHA.dta", replace // dataset of non-hospitalized adults
```

```
use "sepsis.dta", clear // dataset of sepsis hospitalizations
```

*make a variable for the day of sepsis relative to HRS survey, rounded up to the nearest 7

*Note: rounding up here prevents people from being matched to themselves

```
g day=int(sepsis2yday/7)*7+7
```

```
qui append using "masterNHA.dta"
```

ONLINE DATA SUPPLEMENT; Late Mortality after Sepsis; a propensity-matched cohort study

***match sepsis to non-hospitalized adults within days (otherwise will not run)**

```
foreach n of numlist 7 14 to 728 {  
    preserve  
    keep if day==`n'  
    set seed 1234  
    qui cem age gender sepsrisk_percent, treatment(sepsis) k2k  
    keep if cem_matched==1  
    qui save "NHAmatch`n'.dta", replace  
    restore  
}
```

***append day-matches**

```
use "NHAmatch7.dta", clear  
foreach n of numlist 14 21 to 728 {  
    qui append using "NHAmatch`n'.dta"  
}
```